
Zwięzły opis programu PSPICE 
 

 

Opracowanie: Małgorzata Napieralska 

 

Niniejszy dokument został opracowany do celów dydaktycznych  

i moŜna go wykorzystywać wyłącznie do niekomercyjnego uŜytku własnego. 

 

 

  
W celu symulacji układu elektronicznego przy pomocy programu PSPICE naleŜy dokonać edycji 

pliku zawierającego dane o symulowanym obwodzie oraz instrukcje sterujące.  
Nazwa pliku jest dowolna z rozszerzeniem .cir, np. cwiczenie1.cir. Zbiór wyjściowy w postaci 

tekstowej jest automatycznie zakończony .out a zbiór danych graficznych .dat (moŜna go przeglądać przy 
pomocy programu PROBE). Plik moŜe być utworzony przy uŜyciu dowolnego edytora tekstowego np. 
Notepad.  

 
 
 

1.      Opis obwodu 

  

Pierwsza linia zbioru musi zawierać tytuł symulacji np. nazwę symulowanego obwodu. Następnie 
wprowadzany jest opis obwodu z numeracją wszystkich węzłów. Węzeł masy musi mieć numer « 0 ». 
KaŜdy element powinien mieć swoją indywidualną nazwę. Nazwa elementu jest dowolna a pierwsza litera 
oznacza typ elementu: 

  
R - rezystancja 
L - indukcyjność 
K – indukcyjność sprzęŜona 
C - pojemność 
I – niezaleŜne źródło prądowe 
V - niezaleŜne źródło napięciowe 
E - źródło napięcia sterowane napięciem (VCVS) 
H - źródło napięcia sterowane prądem(CCVS) 
G - źródło prądu sterowane napięciem (VCCS) 
F - źródło prądu sterowane prądem (CCCS) 
D - dioda 
Q – tranzystor bipolarny 
J – tranzystor JFET 
M - tranzystor MOS 
X – makromodel (podobwód) 

  
Do opisu kaŜdego elementu przeznaczona jest jedna linia. Po nazwie elementu umieszczamy 

numery węzłów, między którymi znajduje się element; pierwszy węzeł ma potencjał „+” drugi potencjał „-" . 
Następnie umieszczamy wartości elementów w jednostkach SI. MoŜna stosować następujące przedrostki 
: 

  
T = E12             G = E9              MEG = E6         K = E3 
M = E-3             U = E-6             N = E-9             P = E-12           F = E-15 
  
W przypadku elementów półprzewodnikowych  czy makromodeli zamiast wartości podawana jest nazwa 
modelu lub podobwodu. 

WPISYWANIE ELEMENTÓW 
  
Typ elementu Symbol Instrukcja w SPICE 
1)       Rezystancja 
  

R 

 

  
  
RIN3 23 56 1K 

2)       Pojemność 
  

C 

 

  
  
  
C3 12 23 1U 

3)       Indukcyjność 
  

L 

 

  
  
L32 3 4 1N 

 

  
  
VIN 5 6 DC 10 

4)       Źródła niezaleŜne 
  
a) źródło napięciowe 
  

V 
  
  

b) źródło prądowe 
  

I 

 

  
  
IS 6 5 DC 10 

 

  
E1 1 2 4 5 8.0 

5)       Liniowe źródła sterowane 
VCVS - źródło napięcia 
sterowane napięciem E 

  
  
  
  

a)      CCVS – źródło napięcia 
sterowane prądem H 

  
  
  
  

 

  
H1 1 2 VX 8.0 

G1 2 1 4 5 8.0 

  
b)      VCCS – źródło prądu 

sterowane napięciem G 
  
  
  
  
  

c)       CCCS – źródło prądu 
sterowane prądem F 

 

  
F1 2 1 V3 8.0 

6)       dioda 
  

D 

 

  
* Anoda katoda 
D5 23 56 MOD12 

 

7)       Tranzystor bipolarny 
  

Q 

 

  
* C B E 
Q5 32 21 43 BC107  

8)       Tranzystor JFET 
  

J 

 

  
* D G S 
J5 32 21 43 MODJF1  

9)       Tranzystor  MOS 
  

M 

 

  
* D G S Sub. 
M5 32 21 43 55 MOS55  

10)   Makromodel 
(.SUBCKT/.ENDS) 

  
X 

 

  
* IN OUT VDD GND 
X5 55 21 43 0 
INWERTER 
  
  

 

  
  
  
Kolejność umieszczania węzłów przy opisywaniu przyrządów półprzewodnikowych jest następująca: 
  
Dioda: anoda - katoda 
Tranzystor bipolarny: kolektor – baza – emiter (podłoŜe) 
Tranzystor JFET: dren – bramka – źródło  
Tranzystor MOS: dren – bramka – źródło – podłoŜe 
  
W przypadku źródeł sterowanych dwie pierwsze cyfry oznaczają numery węzłów, do których przyłoŜone 
jest źródło. Dla źródła sterowanego napięciowo następne dwie cyfry podają numery węzłów, do których 
przyłoŜone jest napięcie sterujące. Dla źródła sterowanego prądowo naleŜy podać nazwę źródła 
napięciowego, którego prąd jest prądem sterującym. 

Wpisywanie dodatkowych parametrów elementów 
  
Dane elementów mogą być uzupełnione przez podanie dodatkowych informacji za pomocą 
współczynników. 
  
a)      Rezystancja moŜe być zaleŜna od temperatury przez podanie wartości współczynników TC1 i TC2 
  

R(T)=R(Tnom) [ 1 + TC1(δδδδT) + TC2(δδδδT)2] 
  

gdzie: δδδδT= T - Tnom 
  

przykład 
  
R1 2 0 15K TC=0.001 0.0025 
  
b)      Indukcyjność i pojemność mogą być nieliniowe w funkcji prądu lub napięcia co jest opisane przez 

wielomiany 
L=L0 + L1*I + L2*I2 + ... 

C=C0 + C1*U + C2*U2 + ... 
Postać ogólna instrukcji: 

Cxxxxxxx N+ N- POLY C0 C1 C2 ... <IC=INCOND> 
Lxxxxxxx N+ N- POLY L0 L1 L2 ... <IC=INCOND> 

  
Instrukcja  IC=INCOND pozwala na zdefiniowanie warunków początkowych (INITIAL CONDITION) 
napięcia lub prądu niezbędnych do obliczenia stanów początkowych.  
  
przykład 

  
C45 18 21 POLY 2 0.3 IC=10 
  
Gdzie C0=2, C1=0.3, początkowa wartość napięcia na kondensatorze wynosi 10V.  
  
c)       Indukcyjności sprzęŜone opisane są w postaci 
  

Kxxxx  LCxxxx LMxxxx liczba 
przykład 

  
K21 LC1 LMR32 0.99 

  
litera K oznacza, Ŝe indukcyjność LC1 jest sprzęŜona z indukcyjnością LMR32 a współczynnik sprzęŜenia 
wynosi 0.99. 
  
  
d)      Źródła sterowane 
  
  
Mogą być nieliniowe oraz zaleŜeć od wielu zmiennych. Opisuje się je wówczas instrukcją POLY(N), gdzie 
N jest wymiarem zaleŜności. Następnie naleŜy podać N par węzłów dotyczących napięć sterujących lub N 
nazw źródeł napięcia dotyczących prądów sterujących. Na końcu podawane są współczynniki wielomianu 
w następującej postaci dla 2 zmiennych i wielomianu 2go stopnia: 

a0 + a1 x +a2 y +a3 x
2 +a4 xy +a5 y

2 
przykład 

1. dla VCVS: 
E25 7 21 POLY(2) 12 18      31 45   1  10  5  0.5  0.6  0.7   IC=12 
(                              x               y   a0  a1 a2    a3     a4     a5  ) 
2. dla CCCS 
F36 8 54 POLY(2)  VIN        V5   0.1M  1K 10.1  1   1    0.1    IC=1 
(                             x           y         a0  a1  a2     a3     a4     a5  ) 
   
  

2.      Opisywanie źródeł niezaleŜnych 

  

••••        Dziedzina DC  
- podajemy wartość napięcia lub prądu 
  
Przykład 
  
VIN 2 0 DC 10 
IIN 2 0 DC 10 
  

••••        Dziedzina AC  
- podajemy wartość amplitudy i fazy sygnału  
  
  
Przykład 
  
VIN 2 0 AC 1 180 
IIN 2 0 AC 1 0 
  

••••        Dziedzina TR  
  

Istnieje 5 typów źródeł : 
  
a)      Deklaracja źródła impulsowego ma postać 
  
PULSE (V1 V2 TD TR TF PW PER) –  
  
gdzie    V1 – wartość początkowa (w Voltach lub Amperach) 

V2 – wartość impulsu (w Voltach lub Amperach) 
TD – czas opóźnienia (w sekundach, wartość wbudowana 0.0) 
TR – czas narastania (w sekundach, wartość wbudowana 0.0) 
TF – czas opadania (w sekundach, wartość wbudowana 0.0) 
PW – szerokość impulsu 
PER – okres impulsu 

Przykład 
  
Impuls jednostkowy o amplitudzie 1 zapisany jest w postaci : 
  
VUNIT 1 0 PULSE ( 0 1 0 0 ) 
  
b)      Deklaracja źródła sinusoidalnego ma postać 
  
SIN ( VO VA FREQ TD THETA) 
  
Gdzie   VO – wartość składowej stałej (w Woltach lub Amperach) 

VA - amplituda (w Woltach lub Amperach) 
FREQ - częstotliwość - (w Hz, wartość wbudowana f=1/TSTOP) 
TD - czas opóźnienia (w sekundach, wartość wbudowana 0.0) 
THETA – współczynnik w sekundach, wartość wbudowana 0.0) 
  
Źródło sinusoidalne jest opisane wzorem 
  

dla T < TD,  V = VO 
dla TD < T < TSTOP: 
  

V = VO + VA*exp[- (time-TD) *THETA]*sin[2ππππ*FREQ* (time+TD) ] 
  
  
  

Przykład 
  

V31 5 8 SIN ( 0 1 500MEG 0 0) 
  
  
c)       Deklaracja źródła wykładniczego ma postać 
  
EXP ( V1 V2 TD1 TAU1 TD2 TAU2) –  
  
gdzie    V1 – wartość początkowa (w Voltach lub Amperach) 

V2 – wartość impulsu (w Voltach lub Amperach) 
TD1 - czas opóźnienia dla zbocza narastającego impulsu (w sekundach, wartość wbudowana 0.0) 
TAU1 – stała czasu narastania impulsu (w sekundach, wartość wbudowana =TSTOP) 
TD2 – czas opóźnienia określający początek opadania sygnału (w sekundach, wartość 

wbudowana   = TD1+TSTEP) 
TAU2 – stała czasu opadania impulsu (w sekundach, wartość wbudowana =TSTEP) 

Źródło wykładnicze jest opisane wzorem 
  
dla T < TD1, V = V1 
  
dla TD1 < T < TD2, V = V1 + (V2-V1) * [ 1 – exp[-(time-TD1)/TAU1] ] 
  
dla TD2 < T < TSTOP, V = V1 + (V2-V1) * [1 – exp[-(time-TD1)/TAU1] ] + (V1-V2) *  

*[1 – exp[-(time-TD2)/TAU2] ] 
  
  

d)      Deklaracja źródła odcinkowo-liniowego ma postać 
  
PWL (T1 V1 <T2 V2 T3 V3 ...>) –  
  
gdzie źródło zmienia się liniowo i przyjmuje wartości V1 dla czasu T = T1, V2 dla czasu T = T2, itd… 

  

 
  
przykład 
  
dla V1 = 1V, V2 = 1.2V, V3 = 2V, V4 = 2.2V 
i czasów przedstawionych na rysunku : 
  
VIT 1 0 PWL ( 0 1 2U 1 10U 2.2 18U 2.2 21U 1.2 23U 1.2 30U 2) 
  
e)      Deklaracja źródła o modulowanej częstotliwości ma postać 
  
SFFM ( VO VA FC MDI FS) - 
  
gdzie    VO - wartość składowej stałej (w Voltach lub Amperach) 

VA - amplituda (w Voltach lub Amperach) 
FC - częstotliwość nośna - (w Hz, wartość wbudowana f=1/TSTOP) 
MDI - współczynnik modulacji 
FS - częstotliwość sygnału  - (w Hz, wartość wbudowana f=1/TSTOP) 
  

Źródło o modulowanej częstotliwości jest opisane wzorem 
  

V = VO + VA*sin[( 2ππππ*FC*time) + MDI*sin (2 π π π π *FS*time) ] 

przykład 
  
V1 5 8 SFFM ( 0 1M 10K 6 1K ) 
  

3.      Instrukcje sterujące 

  
Po opisie topologii analizowanego obwodu i wpisaniu elementów w programie naleŜy umieścić instrukcje 
sterujące. KaŜda instrukcja sterująca musi zaczynać się od kropki.  
W celu określenia szerokości danych wejściowych i wyjściowych moŜna stosować instrukcję .WIDTH o 
postaci 

.WIDTH IN=80 OUT = 80 
  

a)      Instrukcja  .MODEL 
  
SłuŜy do zdefiniowania modelu przyrządu półprzewodnikowego, elementu nieliniowego lub bramki 
logicznej. Ogólna postać instrukcji jest następująca: 
  
.MODEL               Nazwa modelu              Typ przyrządu                     Nazwa listy parametrów modelu 
  
Nazwa modelu jest dowolna lecz musi zaczynać się od litery. 
  
Typ przyrządu (wybrane elementy) 

D  dioda 

NPN  tranzystor bipolarny NPN 

PNP  tranzystor bipolarny PNP 

NJF  tranzystor  JFET z kanałem typu N 

PJF  tranzystor  JFET z kanałem typu P 

NMOS  tranzystor MOS z kanałem typu N 

PMOS  tranzystor MOS z kanałem typu N 

  

Nazwa listy parametrów modelu zawiera wartości parametrów modelu. Parametry nie określone przez 
uŜytkownika przyjmują wartości wbudowane.  
  
Przykład 
  
.MODEL MOD1 NPN      BF=60 IS=1.2E-13 
  
.MODEL MOD2 D 
  
.MODEL MOD3 NJF       VTO=-3V BETA=1E-4 LAMBDA=2.0E-2 CGS=2PF CGD=2PF 
+                                  PB=1V IS=1.0E-14 
  

b)      Instrukcja  .SUBCKT 
  
Przeznaczona jest do zdefiniowania makromodelu (podobwodu)  (wywoływanego przez  X....). 
Ogólna postać instrukcji jest następująca: 
  
.SUBCKT          Nazwa makromodelu                       Węzły N1 N2 ... 
. 
Opis makromodelu 

. 

. 

.ENDS Nazwa makromodelu 
  
Numeracja węzłów wewnątrz makromodelu jest niezaleŜna od numeracji węzłów obwodu głównego. Nie 
nalęŜy stosować „0”, który jest zarezerwowany dla masy.  
Opis makromodelu  - opisuje się go jak obwód zewnętrzny. Opis kończy instrukcja .ENDS. 
Węzły N1 N2 ... są węzłami zewnętrznymi makromodelu. Nie mogą zawierać węzła o numerze „0”. 
  
Przykład 
  
.SUBCKT WZMOP 1 2 3 4 5 
. 
. 
. 
.ENDS WZMOP 
  
  
W przypadku gdy chcemy stosować makromodel znajdujący się w jednej z bibliotek programu PSPICE 
naleŜy wywołać tę bibliotekę.  
  
Przykład 
  
Stosujemy wzmacniacz operacyjny Ampli-Op µA741 znajdujący się w bibliotece OPNOM.LIB 
  
.LIB OPNOM.LIB 
* wywolanie biblioteki, w której znajduje się Ampli-Op µA741 
X1 1 2 0 4 5 6 UA741 
* umieszczenie wzmacniacza w ukladzie 
  

4.      Analizy w programie PSPICE 

  
Obliczenia wpisanego układu elektronicznego mogą być wykonane w 3 dziedzinach:  

a) analiza statyczna (DC) 
b) analiza prądu przemiennego, zwana teŜ analizą małosygnałową (AC), (dla układu 

zlinearyzowanego w punkcie pracy wyznaczonym w wykonanej uprzednio analizie DC). 
c) analiza stanów przejściowych (TR) 

  
  

a)      Wyznaczenie punktu pracy 
  

Aby otrzymać w zbiorze wyjściowym wszystkie informacje o napięciach, prądach i małosygnałowych 
modelach elementów w punkcie pracy stosujemy instrukcję 
  
.OP 
  

b)      Analiza statyczna 
  

Obliczenia w dziedzinie .DC mogą być wykonane dla wielu wartości prądów i napięć. Instrukcja ma 
postać: 

.DC    Nazwa źródła      Wartość początkowa    Wartość końcowa       Krok 
  
W przypadku wielu źródeł naleŜy wpisać kolejno wartości ich parametrów 
  
Przykład 
  
.DC VCE 0 100 5 VBE 0 0.7 0.1 
  

c)      Analiza małosygnałowa 
  

Zmienną niezaleŜną w analizie .AC jest częstotliwość. Zmienia się ona od wartości FSTART do FSTOP. 
Dla analizy .AC naleŜy zdefiniować minimum jedno źródło AC podając jego amplitudę i fazę. Postacie 
analizy : 
  
.AC DEC ND FSTART FSTOP 
.AC OCT NO FSTART FSTOP 
.AC LIN NP FSTART FSTOP 
  
gdzie    DEC: skala logarytmiczna, ND liczba punktów na dekadę. 

OCT: skala logarytmiczna, NO liczba punktów na oktawę. 
LIN: zmiany liniowe, NP liczba punktów między FSTART i FSTOP. 

  
Przykład 
  
.AC DEC 5 1K 10MEG 
  
  

W celu otrzymania charakterystyki przejściowej (output/input), rezystancji wejściowej i wyjściowej stosuje 
się instrukcję .TF. Pierwsza zmienna następująca po tej instrukcji określa zmienną wyjściową, druga 
wejściową. 
  
Przykład 
  
.TF V(23) VIN1 
  
PSPICE liczy stosunek V(23)/VIN, małosygnałową rezystancję wejściową dla VIN1 oraz małosygnałową 
rezystancję wyjściową dla V(23). 
  

d)      Analiza stanów przejściowych 
  
Wywoływana jest przy pomocy instrukcji .TRAN. Zmienną niezaleŜną w analizie jest czas.  
Postać analizy: 
  
.TRAN  TSTEP  TSTOP  <TSTART  <TMAX>>   <UIC> 
  
gdzie 

TSTEP    - wartość kroku dla wydruku wyników 
TSTOP   - czas końcowy symulacji 
TSTART  - początek symulacji (wartość wbudowana =0) 
TMAX     - maksymalna wartość kroku (wartość wbudowana TSTOP/50) 
UIC         - polecenie uwzględnienia warunków początkowych określonych w instrukcji .IC i nie 

wykonywania analizy DC 

Przykład 
  
.TRAN 1N 0.2U UIC 

5.      Wyniki obliczeń 

  
Wyniki moŜna przedstawić w formie: 
a) .PRINT – forma tabelaryczna 
b) .PLOT – rysowanie w formie alfanumerycznej 
c) .PROBE – forma graficzna. Zastosowanie postprocesora graficznego pozwala na wszechstronne 
opracowanie otrzymanych wykresów. 

Funkcje sygnałów dostępne przy uŜyciu  .PROBE 
  
ABS(x)  wartość bezwzględna 
SGN(x)  1 dla x > 0, 0 dla x=0, -1 dla x < 0 
SQRT(x)  pierwiastek kwadratowy 
EXP(x)   funkcja wykładnicza 
LOG(x)  logarytm naturalny 
LOG10(x) logarytm décimal 
DB(x)     decybel (20 log2(x)) 
PWR(x,y) potęga (abs(x))y 
SIN(x)    sinus (x w radianach) 
COS(x)  cosinus 
TAN(x)   tangens 
ATAN(x) arctangens 
d(y)       pochodna y względem zmiennej osi x 
s(y)        całka y względem zmiennej osi x 
AVG(y)  wartość średnia y 
RMS(y)  wartość skuteczna y 
  

6.      Wybrane analizy dodatkowe 

a)      Zmiana temperatury 
  
Obliczenia wykonywane są dla temperatury nominalnej TNOM=27 0C (300K). MoŜna ją zmienić globalnie 
dla całego symulowanego układu  przy uŜyciu następującej instrukcji: 
  
.TEMP = wartość 
 ( TEMP > -223oC ) 
  
Przykład 
  
.TEMP = 125 
  
Dla elementów, które zaleŜą od temperatury naleŜy podać odpowiednie współczynniki temperaturowe. 
Temperatura odniesienia pozostaje równa TNOM = 27oC. W celu zmiany temperatury odniesienia naleŜy 
zastosować instrukcję .OPTIONS. 
  
Przykład 
  
.OPTIONS TNOM = wartosc 



 12

  
  
Instrukcji tej moŜna uŜyć w celu wyeliminowania stronicowania i drukowania nagłówków na kaŜdej stronie, 
co jest wygodne gdy pliki wyjściowe oglądamy na ekranie monitora. 
  
.OPTIONS NOPAGE 

b)      Obliczanie wraŜliwości 
WraŜliwość (małosygnałowa AC) jest liczona dla kaŜdej  podanej zmiennej wyjściowej względem 
wszystkich parametrów obwodu. 
Ogólna postać instrukcji: 
  
.SENS OV1 <OV2 ...> 
  
Przykład 
  
 .SENS V(1) V(2,6) I(VIN) 
  

c)      Analiza Fouriera 
  

Wykonywana jest dla analizy czasowej TRAN 
Ogólna postać instrukcji: 
  
.FOUR FREQ OV1 <OV2 OV3 ...> 
gdzie    FREQ               częstotliwość sygnału 

OV1 OV2 OV3 ...  sygnały, które mają być poddane analizie Fouriera 
  

Przykład 
  
.FOUR 10K V(1) 
  

d)      Definiowanie punktów startowych i warunków początkowych 
  
Instrukcja .NODESET stosowana jest dla analizy DC i ustawia punkt startowy do nieliniowych iteracji. 
Przydatna np. przy analizie układów przerzutnikowych w celu wymuszenia ich stanu początkowego. 
Ogólna postać instrukcji: 
  
.NODESET     V(NODNUM)=WARTOŚĆ      V(NODNUM)= WARTOŚĆ... 
  
Przykład 
  
.NODESET    V(1)=10    V(2)=5.1     V(14)=7.12 
  
  
Warunki początkowe dla analizy TRAN określa się za pomocą instrukcji .IC 
Ogólna postać instrukcji: 
  
.IC     V(NODNUM)= WARTOŚĆ       V(NODNUM)= WARTOŚĆ ... 
  
Przykład 
  
.IC    V(4)=14    V(3)=6.9     V(24)=33 

e)      Analiza szumowa 
  
Wykonywana jest dla analizy zmiennoprądowej AC i powinna być poprzedzona analizą AC. Składowe 
widma mocy szumu, pochodzące od kaŜdego z elementów układu, są sumowane na wyjściu 
napięciowym. Do przeprowadzenia analizy szumowej potrzebne są modele szumowe elementów. W 
układzie mogą występować dwa rodzaje elementów szumowych: rezystory generujące szumy termiczne, 
przyrządy półprzewodnikowe generujące szumy termiczne i szumy migotania. 
  
  
  
Ogólna postać instrukcji: 
  
 .NOISE    OUTV    INSRC     NUMS 
gdzie    OUTV   napięcie wyjściowe, dla którego jest analizowany wkład kaŜdej składowej szumu  

INSRC źródło niezaleŜnego napięcia lub prądu, dla którego jest liczony równowaŜny szum 
wejściowy 
NUMS liczba określająca co ile punktów na osi częstotliwości (w stosunku do analizy 
zmiennoprądowej) będzie wykonywana analiza szumów. 
  

Przykład 
  

.NOISE V(3) VIN 5 
  

f)       Analiza Monte Carlo 
  

UmoŜliwia wykonanie obliczeń z uwzględnieniem tolerancji wartości parametrów modeli. Podajemy ile 
razy ma być wykonana analiza, kaŜdorazowo po wyznaczeniu nowych wartości zmienianych parametrów 
oraz rodzaj analizy głównej (DC, AC, TRAN). Obliczenia analizy Monte Carlo są wykonywane w ramach 
odrębnego typu analizy i nie mają wpływu na wyniki pozostałych analiz. 
  
Ogólna postać instrukcji: 
  
.MC  RUNS ANALYSIS VOUT FUN options 
  
gdzie    RUNS - liczba przeprowadzonych analiz Monte Carlo 
            ANALYSIS - rodzaj analizy, np. AC, DC, TRAN 
            VOUT - zmienna obwodowa, którą chcemy badać 

FUN - funkcja obliczana na podstawie wartości wszystkich przebiegów analizy Monte Carlo w celu 
określenia liczbowego parametru określającego wraŜliwość zmiennej VOUT na przypadkowe 
zmiany parametrów modeli. Najczęściej stosuje się YMAX oznaczającą maksymalną 
odchyłkę zmiennej od wartości nominalnej. 

options - jedna lub więcej opcji analizy Monte Carlo. Najczęściej stosowane opcje to list 
oznaczająca wypisanie parametrów poszczególnych modeli w kaŜdym przebiegu analizy i 
output all po zastosowaniu której wyniki wszystkich analiz moŜemy porównać w programie 
probe na jednym wykresie. 

  

g)      Analiza najgorszego przypadku (Worst Case) 
  

W przeciwieństwie do analizy Monte Carlo analiza najgorszego przypadku uwzględnia jedynie skrajne 
wartości tolerancji parametrów przy wykonywaniu obliczeń układu. 
  
Ogólna postać instrukcji: 

.WCASE  ANALYSIS VOUT FUN 
  
oznaczenia identyczne jak w przypadku analizy Monte Carlo. 
  
  
Przykład programu do analizy wzmacniacza w programie PSPICE (po lewej) i jego modyfikacja w celu 
przeprowadzenia analizy Monte Carlo (po prawej) 
  
  
WZMACNIACZ 
.OPTIONS ACCT LIST NOPAGE 
.WIDTH IN=80 OUT=132 
.MODEL MODTR1 NPN 
.MODEL MODTR2 PNP 
RG 1 2 1.5K 
RB 4 3 18K 
R1 4 7 85 
R2 7 6 17 
R3 6 5 26 
RL 9 0 18 
C1 2 3 100U 
C2 8 9 100U 
QTA 5 3 0 MODTR1 
QTB 7 6 5 MODTR1 
QTC 4 7 8 MODTR1 
QTD 0 5 8 MODTR2 
VC 4 0 DC 12V 
VIN 1 0 AC 1 0 SIN(0 1 1E3 1M) 
.OP 
.TRAN 0.05M 4M 
.FOUR 1K V(9) 
.PLOT TRAN V(9) 
.AC DEC 10 100HZ 100KHZ 
.PLOT AC VM(9) VP(9) 
.PROBE 
.END 
  

WZMACNIACZ 
.OPTIONS ACCT LIST NOPAGE 
.MODEL MODTR1 NPN 
.MODEL MODTR2 PNP 
.MODEL R1 RES (R=1 DEV 10%)  
.MODEL C1 CAP (C=1 DEV 10%)  
RG 1 2 R1 1.5K 
RB 4 3 R1 18K 
R1 4 7 R1 85 
R2 7 6 R1 17 
R3 6 5 R1 26 
RL 9 0 R1 18 
C1 2 3 C1 100U 
C2 8 9 C1 100U 
QTA 5 3 0 MODTR1 
QTB 7 6 5 MODTR1 
QTC 4 7 8 MODTR1 
QTD 0 5 8 MODTR2 
VC 4 0 DC 12V 
VIN 1 0 AC 1 0 SIN(0 1 1E3 1M) 
.TRAN 0.05M 4M 
.MC 100 TRAN V(9) YMAX LIST OUTPUT ALL 
*.WCASE TRAN V(9) YMAX 
.PROBE 
.END 

